Gosper

In[1]:=
```
<< RISC`fastZeil`
```

Fast Zeilberger Package version 3.61
written by Peter Paule, Markus Schorn, and Axel Riese

In[126]:=
```
Gosper[(4 k + 1) * k! / (2 k + 1)!, k]
```

Out[126]= $\left\{\frac{(1+4\,k)\ k!}{(1+2\,k)\ !} == \Delta_k\left[-\frac{2\ (1+2\,k)\ k!}{(1+2\,k)\ !}\right]\right\}$

In[127]:=
```
Gosper[k!, k]
```

Out[127]= { }

In[128]:=
```
Gosper[Binomial[n, k], k]
```

Out[128]= { }

In[129]:=
```
Gosper[(2 k - n - 1) / (n - k + 1) * Binomial[n, k], k]
```

Out[129]= $\left\{-\frac{(-1+2\,k-n)\ \text{Binomial}[n,\,k]}{-1+k-n} == \Delta_k\left[\frac{k\ \text{Binomial}[n,\,k]}{-1+k-n}\right]\right\}$

Zeilberger

```
In[130]:= Zb[Binomial[n, k], k, n]
Out[130]= {2 F[k, n] - F[k, 1 + n] == Δk[F[k, n] R[k, n]]}

In[131]:= show[R]
Out[131]= k/(1 - k + n)

In[132]:= Zb[(-1)^k * Binomial[2 n, n + k]^2, k, n]
Out[132]= {-2 (1 + 2 n) F[k, n] + (1 + n) F[k, 1 + n] == Δk[F[k, n] R[k, n]]}

In[133]:= Zb[Binomial[n, k]^2 * Binomial[n + k, k]^2, k, n]
Out[133]= {(1 + n)^3 F[k, n] - (3 + 2 n) (39 + 51 n + 17 n^2) F[k, 1 + n] + (2 + n)^3 F[k, 2 + n] ==
            Δk[F[k, n] R[k, n]]}

In[134]:= Zb[(-1)^k * Binomial[n, k] * Binomial[2 * k, n], k, n]
Out[134]= {-2 (1 + n) F[k, n] + (-1 - n) F[k, 1 + n] == Δk[F[k, n] R[k, n]]}

In[135]:= Zb[(-1)^k * Binomial[n, k] * Binomial[3 * k, n], k, n]
Out[135]= {9 (1 + n) (2 + n) F[k, n] + 3 (2 + n) (7 + 5 n) F[k, 1 + n] + 2 (2 + n) (3 + 2 n) F[k, 2 + n] ==
            Δk[F[k, n] R[k, n]]}

In[136]:= Zb[(-1)^k * Binomial[n, k] * Binomial[4 * k, n], k, n]
Out[136]= {-64 (1 + n) (2 + n) (3 + n) (7 + 3 n) F[k, n] - 16 (2 + n) (3 + n) (107 + 125 n + 33 n^2) F[k, 1 + n] -
             4 (3 + n) (4 + 3 n) (218 + 180 n + 37 n^2) F[k, 2 + n] -
             3 (3 + n) (4 + 3 n) (7 + 3 n) (8 + 3 n) F[k, 3 + n] == Δk[F[k, n] R[k, n]]}

In[137]:= Zb[Pochhammer[a, k] * Pochhammer[b, k] / Pochhammer[c, k] / k! * z^k, k, c]
Out[137]= {c (1 + c) (-1 + z) F[k, c] - (1 + c) (-c + z - a z - b z + 2 c z) F[k, 1 + c] +
              (-1 + a - c) (-1 + b - c) z F[k, 2 + c] == Δk[F[k, c] R[k, c]]}
```

Univariate D-finite functions

In[2]:=
```
<< RISC`HolonomicFunctions`
```

HolonomicFunctions Package version 1.7.3 (21-Mar-2017)
written by Christoph Koutschan
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

```
--> Type  ?HolonomicFunctions  for help.
```

In[138]:=
```
Annihilator[Erf[1/(x^2+1)] * Exp[1/(x^2+1)], Der[x]]
```

Out[138]= $\{(x + 5 x^3 + 10 x^5 + 10 x^7 + 5 x^9 + x^{11}) D_x^2 + (-1 - x^2 + 10 x^4 + 22 x^6 + 15 x^8 + 3 x^{10}) D_x + (-4 x^3 + 4 x^5)\}$

In[139]:=
```
Annihilator[Sinh[x]^2 + Cosh[x]^(-2), Der[x]]
```

Annihilator: The expression Sech[x] is not recognized to be ∂–finite. The result might not generate a zero–dimensional ideal.

Out[139]= { }

In[140]:=
```
Annihilator[Log[Sqrt[1-x]] / Exp[Sqrt[1-x]], Der[x]]
```

Out[140]= $\{(80 - 304 x + 432 x^2 - 272 x^3 + 64 x^4) D_x^4 + (-336 + 928 x - 848 x^2 + 256 x^3) D_x^3 + (156 - 212 x + 24 x^2 + 32 x^3) D_x^2 + (48 - 84 x + 32 x^2) D_x + (7 - 13 x + 4 x^2)\}$

In[141]:=
```
Annihilator[ArcTan[Exp[x]], Der[x]]
```

DFiniteSubstitute: The substitutions for continuous variables $\{e^x\}$ are supposed to be algebraic expressions. Not all of them are recognized to be algebraic. The result might not generate a ∂–finite ideal.

Annihilator: The expression (w.r.t. {Der[x]}) is not recognized to be ∂–finite. The result might not generate a zero–dimensional ideal.

Out[141]= { }

In[142]:=
```
Annihilator[Exp[ArcTan[x]], Der[x]]
```

Out[142]= $\{(-1 - x^2) D_x + 1\}$

Annihilator of LegendreP

In[3]:=
```
{(x^2-1) * D[P[n, x], x, x] + 2 x * D[P[n, x], x] - n (n+1) P[n, x] == 0,
 n * P[n, x] == (2 n - 1) x * P[n - 1, x] - (n - 1) * P[n - 2, x]}
```

Out[3]= $\{-n (1 + n) P[n, x] + 2 x P^{(0,1)}[n, x] + (-1 + x^2) P^{(0,2)}[n, x] == 0, n P[n, x] == -(-1 + n) P[-2 + n, x] + (-1 + 2 n) x P[-1 + n, x]\}$

In[4]:=
```
ToOrePolynomial[%, P[n, x]]
```

Out[4]= $\{(-1 + x^2) D_x^2 + 2 x D_x + (-n - n^2), (2 + n) S_n^2 + (-3 x - 2 n x) S_n + (1 + n)\}$

In[5]:=
```
OreGroebnerBasis[%]
```

Out[5]= $\{(-1-n)\,S_n + (-1+x^2)\,D_x + (x+n\,x),\ (-1+x^2)\,D_x^2 + 2\,x\,D_x + (-n-n^2)\}$

Ebisu

In[6]:=
```
ann2F1 = Annihilator[Hypergeometric2F1[a, b, c, z], {S[a], S[b], S[c], Der[z]}]
```

Out[6]= $\{(a\,b - a\,c - b\,c + c^2)\,S_c + (-c + c\,z)\,D_z + (a\,c + b\,c - c^2),$
$b\,S_b - z\,D_z - b,\ a\,S_a - z\,D_z - a,\ (-z + z^2)\,D_z^2 + (-c + z + a\,z + b\,z)\,D_z + a\,b\}$

In[7]:=
```
rhs = (16 / 27) ^t * Gamma[t + 5 / 6] * Gamma[2 / 3] / Gamma[t + 2 / 3] / Gamma[5 / 6]
```

Out[7]= $$\frac{\left(\frac{16}{27}\right)^t \,\text{Gamma}\left[\frac{2}{3}\right]\,\text{Gamma}\left[\frac{5}{6}+t\right]}{\text{Gamma}\left[\frac{5}{6}\right]\,\text{Gamma}\left[\frac{2}{3}+t\right]}$$

In[8]:=
```
DFiniteSubstitute[ann2F1,
  {a → 2 t, b → 2 t + 1 / 3, c → t + 5 / 6, z → -1 / 8}, Algebra → OreAlgebra[S[t]]]
```

Out[8]= $\{(54 + 81\,t)\,S_t + (-40 - 48\,t)\}$

In[9]:=
```
Annihilator[Hypergeometric2F1[2 t, 2 t + 1 / 3, t + 5 / 6, -1 / 8], S[t]]
```

Out[9]= $\{(54 + 81\,t)\,S_t + (-40 - 48\,t)\}$

In[10]:=
```
ApplyOreOperator[%, rhs] // FunctionExpand
```

Out[10]= $\{0\}$

In[11]:=
```
rel = FindRelation[ann2F1, Support → {S[a] ^2 * S[b] ^2 * S[c], Der[z], 1}][[1]]
```

Out[11]= $(a\,b + a^2\,b + a\,b^2 + a^2\,b^2 - 2\,a\,b\,z - 2\,a^2\,b\,z - 2\,a\,b^2\,z - 2\,a^2\,b^2\,z + a\,b\,z^2 + a^2\,b\,z^2 + a\,b^2\,z^2 + a^2\,b^2\,z^2)$
$S_a^2\,S_b^2\,S_c + (-c - a\,c - b\,c - a\,b\,c + 2\,c^2 + a\,c^2 + b\,c^2 - c^3 - c\,z - 2\,a\,c\,z - a^2\,c\,z - 2\,b\,c\,z -$
$a\,b\,c\,z - b^2\,c\,z + c^2\,z + a\,c^2\,z + b\,c^2\,z)\,D_z + (-2\,a\,b\,c - a^2\,b\,c - a\,b^2\,c + a\,b\,c^2)$

In[12]:=
```
cf = Coefficient[rel, Der[z]]
```

Out[12]= $-c - a\,c - b\,c - a\,b\,c + 2\,c^2 + a\,c^2 + b\,c^2 - c^3 - c\,z -$
$2\,a\,c\,z - a^2\,c\,z - 2\,b\,c\,z - a\,b\,c\,z - b^2\,c\,z + c^2\,z + a\,c^2\,z + b\,c^2\,z$

In[13]:=
```
Expand[cf /. {a → a + 2 t, b → b + 2 t, c → c + t}]
```

Out[13]= $-c - a\,c - b\,c - a\,b\,c + 2\,c^2 + a\,c^2 + b\,c^2 - c^3 - t - a\,t - b\,t - a\,b\,t + c^2\,t - 2\,t^2 - a\,t^2 - b\,t^2 + c\,t^2 - t^3 -$
$c\,z - 2\,a\,c\,z - a^2\,c\,z - 2\,b\,c\,z - a\,b\,c\,z - b^2\,c\,z + c^2\,z + a\,c^2\,z + b\,c^2\,z - t\,z - 2\,a\,t\,z - a^2\,t\,z - 2\,b\,t\,z -$
$a\,b\,t\,z - b^2\,t\,z - 6\,c\,t\,z - 4\,a\,c\,t\,z - 4\,b\,c\,t\,z + 4\,c^2\,t\,z - 7\,t^2\,z - 5\,a\,t^2\,z - 5\,b\,t^2\,z - 4\,c\,t^2\,z - 8\,t^3\,z$

In[14]:=
```
Solve[CoefficientList[%, t] == 0, {a, b, c, z}]
```

Solve: Equations may not give solutions for all "solve" variables.

Out[14]=

$$\left\{\left\{b\to\frac{1}{3}(-1+3a),\ c\to\frac{1}{6}(4+3a),\ z\to-\frac{1}{8}\right\},\ \left\{b\to\frac{1}{3}(1+3a),\ c\to\frac{1}{6}(5+3a),\ z\to-\frac{1}{8}\right\},\ \left\{a\to-\frac{5}{3},\ b\to-\frac{4}{3},\ c\to 0,\ z\to-\frac{1}{8}\right\},\ \left\{a\to-\frac{4}{3},\ b\to-\frac{5}{3},\ c\to 0,\ z\to-\frac{1}{8}\right\}\right\}$$

In[15]:=
```
Take[%, 2] /. a → 0
```

Out[15]=

$$\left\{\left\{b\to-\frac{1}{3},\ c\to\frac{2}{3},\ z\to-\frac{1}{8}\right\},\ \left\{b\to\frac{1}{3},\ c\to\frac{5}{6},\ z\to-\frac{1}{8}\right\}\right\}$$

In[16]:=
```
ApplyOreOperator[rel, F[a, b, c, z]]
```

Out[16]=

$$\left(-2abc-a^2bc-ab^2c+abc^2\right)F[a,b,c,z]+\left(ab+a^2b+ab^2+a^2b^2-2abz-2a^2bz-2ab^2z-2a^2b^2z+abz^2+a^2bz^2+ab^2z^2+a^2b^2z^2\right)F[2+a,2+b,1+c,z]+\left(-c-ac-bc-abc+2c^2+ac^2+bc^2-c^3-cz-2acz-a^2cz-2bcz-abcz-b^2cz+c^2z+ac^2z+bc^2z\right)F^{(0,0,0,1)}[a,b,c,z]$$

In[17]:=
```
Simplify[% /. {a → 2 t, b → 2 t + 1/3, c → t + 5/6, z → -1/8}]
```

Out[17]=

$$-\frac{1}{48}t\left(1+8t+12t^2\right)\left(8(5+6t)F\left[2t,\frac{1}{3}+2t,\frac{5}{6}+t,-\frac{1}{8}\right]-27(2+3t)F\left[2+2t,\frac{7}{3}+2t,\frac{11}{6}+t,-\frac{1}{8}\right]\right)$$

Finite Element Methods

In[121]:=
```
annphi = Annihilator[(1 - x) ^ i * JacobiP[j, 2 i + 1, 0, 2 x - 1] *
    LegendreP[i, 2 y / (1 - x) - 1], {S[i], S[j], Der[x], Der[y]}]
```

Out[121]=

$$\{(2+2i+3j+2ij+j^2)S_j+(3x+2ix+2jx-3x^2-2ix^2-2jx^2)D_x+(-3xy-2ixy-2jxy)D_y+(2+2i+3j+2ij+j^2-6x-7ix-2i^2x-7jx-4ijx-2j^2x),\ \langle\langle 1\rangle\rangle,\ \langle\langle 4\rangle\rangle,\ (\langle\langle 190\rangle\rangle+62j^4x^3+129ij^4x^3+85i^2j^4x^3+18i^3j^4x^3+4j^5x^3+6ij^5x^3+2i^2j^5x^3)S_i^2+\langle\langle 3\rangle\rangle+(\langle\langle 524\rangle\rangle+\langle\langle 1\rangle\rangle+40ij^5xy^2+16i^2j^5xy^2)\}$$

large output | show less | show more | show all | set size limit...

In[122]:=
```
ByteCount[annphi]
```

Out[122]= 461 272

In[123]:=
```
Support[annphi]
```

Out[123]=

$$\{\{S_j, D_x, D_y, 1\},\ \{D_y^2, D_y, 1\},\ \{D_x D_y, S_i, D_x, D_y, 1\},\ \{D_x^2, S_i, D_x, D_y, 1\},\ \{S_i D_y, S_i, D_x, D_y, 1\},\ \{S_i D_x, S_i, D_x, D_y, 1\},\ \{S_i^2, S_i, D_x, D_y, 1\}\}$$

In[124]:=
```
FindRelation[annphi, Eliminate → {x, y}, Pattern → {_, _, 0 | 1, 0}]
```

Out[124]=

$$\{(-25 - 20 i - 4 i^2 - 15 j - 6 i j - 2 j^2) S_i S_j^2 D_x + (-15 - 6 i - 11 j - 2 i j - 2 j^2) S_j^3 D_x + (-18 - 18 i - 4 i^2 - 6 j - 4 i j) S_i S_j D_x + (6 + 14 i + 4 i^2 + 2 j + 4 i j) S_j^2 D_x + (210 + 214 i + 72 i^2 + 8 i^3 + 214 j + 144 i j + 24 i^2 j + 72 j^2 + 24 i j^2 + 8 j^3) S_i S_j + (7 + 2 i + 9 j + 2 i j + 2 j^2) S_i D_x + (210 + 214 i + 72 i^2 + 8 i^3 + 214 j + 144 i j + 24 i^2 j + 72 j^2 + 24 i j^2 + 8 j^3) S_j^2 + (21 + 20 i + 4 i^2 + 13 j + 6 i j + 2 j^2) S_j D_x\}$$

In[125]:=
```
ApplyOreOperator[Factor[First[%]], ϕi,j[x]]
```

Out[125]=

$$2 (3 + i + j) (5 + 2 i + 2 j) (7 + 2 i + 2 j) \phi_{i,2+j}[x] + 2 (3 + i + j) (5 + 2 i + 2 j) (7 + 2 i + 2 j) \phi_{1+i,1+j}[x] + (3 + 2 i + j) (7 + 2 i + 2 j) \phi_{i,1+j}'[x] + 2 (1 + 2 i) (3 + i + j) \phi_{i,2+j}'[x] - (3 + j) (5 + 2 i + 2 j) \phi_{i,3+j}'[x] + (1 + j) (7 + 2 i + 2 j) \phi_{1+i,j}'[x] - 2 (3 + 2 i) (3 + i + j) \phi_{1+i,1+j}'[x] - (5 + 2 i + j) (5 + 2 i + 2 j) \phi_{1+i,2+j}'[x]$$

Relativistic Coulomb Integrals

In[18]:=
```
expr = LaguerreL[-1 + n, ν, r]^2 +
    LaguerreL[-1 + n, ν, r] LaguerreL[n, ν, r] + LaguerreL[n, ν, r]^2;
```

In[19]:=
```
Timing[
 ann1 = Annihilator[expr, {S[n], Der[r], S[p]}];
 ByteCount[ann1]
]
```

Out[19]= {5.01859, 2 354 144}

In[20]:=
```
UnderTheStaircase[ann1]
```

Out[20]= $\{1, D_r, S_n, D_r^2, S_n D_r, S_n^2, D_r^3, S_n D_r^2, S_n^2 D_r, S_n^3\}$

In[21]:=
```
Length[%]
```

Out[21]= 10

In[22]:=
```
op = 1 + S[n] + S[m] ** S[n];
expr = ApplyOreOperator[op, LaguerreL[m - 1, ν, r] * LaguerreL[n - 1, ν, r]] /. m → n
```

Out[23]= LaguerreL[-1 + n, ν, r]^2 + LaguerreL[-1 + n, ν, r] LaguerreL[n, ν, r] + LaguerreL[n, ν, r]^2

In[24]:=
```
annLL = Annihilator[LaguerreL[m - 1, ν, r] * LaguerreL[n - 1, ν, r], {S[m], S[n], Der[r]}]
```

Out[24]=

$$\{m S_m + n S_n - r D_r + (-m - n + 2 r - 2 \nu),\ 2 n r S_n D_r - r^2 D_r^2 + (-2 n r + 2 n \nu) S_n + (-r - 2 n r + 3 r^2 - 3 r \nu) D_r + (2 r - m r + 3 n r - 2 r^2 - 2 n \nu + 4 r \nu - 2 \nu^2),\ (1 + n) S_n^2 + (-1 - 2 n + r - \nu) S_n + (n + \nu),\ r^2 D_r^3 + (3 r - 3 r^2 + 3 r \nu) D_r^2 + (2 m n - 2 n^2) S_n + (1 - 8 r + m r + 3 n r + 2 r^2 + 3 \nu - 4 r \nu + 2 \nu^2) D_r + (-2 + m + n - 2 m n + 2 n^2 + 4 r - 4 n r - 4 \nu + 4 n \nu)\}$$

In[25]:=
```
Timing[
 ann2 = DFiniteOreAction[annLL, op];
 ByteCount[ann2]
]
```

Out[25]= {0.736842, 300280}

In[26]:=
```
Timing[
 annF2G2 = DFiniteSubstitute[ann2, {m → n}];
 ByteCount[annF2G2]
]
```

Out[26]= {1.39144, 171456}

In[27]:=
```
Support[annF2G2]
```

Out[27]= $\{\{S_n D_r, D_r^2, S_n, D_r, 1\}, \{S_n^2, D_r^2, S_n, D_r, 1\}, \{D_r^3, D_r^2, S_n, D_r, 1\}\}$

In[28]:=
```
UnderTheStaircase[annF2G2]
```

Out[28]= $\{1, D_r, S_n, D_r^2\}$

Example from Gradshteyn & Ryzhik

In[29]:=
```
Annihilator[Pi * 2 ^ (1 - nu) * I ^ n *
   Gamma[2 nu + n] / n ! / Gamma[nu] * a ^ (-nu) * BesselJ[nu + n, a], {S[n], Der[a]}]
```

Out[29]= $\{(a + a\,n)\,S_n + (i\,a\,n + 2\,i\,a\,nu)\,D_a + (-i\,n^2 - 2\,i\,n\,nu),\ a^2 D_a^2 + (a + 2\,a\,nu)\,D_a + (a^2 - n^2 - 2\,n\,nu)\}$

In[30]:=
```
CreativeTelescoping[(1 - x ^ 2) ^ (nu - 1 / 2) * Exp[I * a * x] * GegenbauerC[n, nu, x],
   Der[x], {S[n], Der[a]}] // Timing
```

Out[30]= $\{0.307348, \{\{(a + a\,n)\,S_n + (i\,a\,n + 2\,i\,a\,nu)\,D_a + (-i\,n^2 - 2\,i\,n\,nu),$
$a^2 D_a^2 + (a + 2\,a\,nu)\,D_a + (a^2 - n^2 - 2\,n\,nu)\},$
$\{i\,(1 + n)\,S_n - i\,(n\,x + 2\,nu\,x),\ (1 + n)\,S_n - i\,(-a - i\,n\,x - 2\,i\,nu\,x + a\,x^2)\}\}\}$

Zeilberger's Problem

In[31]:=
```
(* This is the identity we wish to show,
where Dp(x) is defined by the second-order recurrence below. *)
TraditionalForm[
 Dp[x] == R1[p] + Sum[R2[i, p], {i, 1, p - 1}] + Sum[R3[i, p] * Di[x], {i, 1, p - 1}]]
```

Out[31]//TraditionalForm=

$$D_p(x) = \sum_{i=1}^{p-1} D_i(x)\,\mathrm{R3}(i, p) + \sum_{i=1}^{p-1} \mathrm{R2}(i, p) + \mathrm{R1}(p)$$

In[32]:=
```
(* This is the recursive definition of D_p(x) *)
R1[p_] := (-12 * p^2 * (x - p) / (x^3 - x + p - p^3)) * (-28 / 9 * x^2 + 29 / 45 + 274 / 45 * p^2 -
    1 / 6 * (x^3 - x) / p + 1 / 5 * (x^3 - x) * (x + p) / (x^2 + x * p + p^2 - 1) - 13 / 9 * x * p);
R2[i_, p_] := (12 * i^2 * (i - x) / x / (x + 1) / (x - 1)) *
   ((-12) * p * (p - i) * (x - p) / (x - i) / (x^3 - x + p - p^3)) *
   (5 / 18 * (x^3 - x) * 1 / p + 38 / 15 * p^2 + (x^3 - x) / 5 * (x + p) / (x^2 + p^2 + x * p - 1) -
    13 / 9 * i * p - 13 / 9 * (p - i) * x + 49 / 45);
R3[i_, p_] := (-12) * p * (p - i) * (x - p) / (x - i) / (x^3 - x + p - p^3);
Clear[DxR];
DxR[p_, x_] := DxR[p, x] =
  If[p == 1,
   Together[(-12 * (x - 1) / (x^3 - x) *
      (-28 / 9 * x^2 + 29 / 45 + 274 / 45 - 1 / 6 * (x^3 - x) + 1 / 5 * (x^2 - 1) - 13 / 9 * x))],
   Together[R1[p] + Sum[R2[i, p], {i, 1, p - 1}] +
     Sum[R3[i, p] * DxR[i, x], {i, 1, p - 1}]]
  ];
Table[DxR[n, x], {n, 6}]
```

Out[37]=

$$\Big\{\frac{2\left(-588+115x+262x^2+15x^3\right)}{15x(1+x)},$$
$$\left(4\left(16056+1266x-3649x^2-4910x^3+490x^4+524x^5+15x^6\right)\right)/\left(15(-1+x)x(1+x)\left(3+2x+x^2\right)\right),$$
$$\left(6\left(-372672-48120x+44530x^2+112525x^3-1642x^4-8625x^5-5422x^6+375x^7+262x^8+5x^9\right)\right)/\left(5(-2+x)(-1+x)x(1+x)\left(3+2x+x^2\right)\left(8+3x+x^2\right)\right),$$
$$\left(8\left(149921280+22889088x-10710360x^2-43049898x^3-828625x^4+2305958x^5+2582700x^6-79350x^7-100470x^8-38046x^9+2020x^{10}+1048x^{11}+15x^{12}\right)\right)/\left(15(-3+x)(-2+x)(-1+x)x(1+x)\left(3+2x+x^2\right)\left(8+3x+x^2\right)\left(15+4x+x^2\right)\right),$$
$$\left(10\left(-6830438400-1146908160x+315279648x^2+1890217728x^3+73433634x^4-73063357x^5-120642254x^6+1284739x^7+3665192x^8+2446002x^9-80136x^{10}-57590x^{11}-14746x^{12}+635x^{13}+262x^{14}+3x^{15}\right)\right)/\left(3(-4+x)(-3+x)(-2+x)(-1+x)x(1+x)\left(3+2x+x^2\right)\left(8+3x+x^2\right)\left(15+4x+x^2\right)\left(24+5x+x^2\right)\right),$$
$$\left(12\left(4066288128000+727574400000x-126670875840x^2-1097400884256x^3-57311836140x^4+31601361388x^5+71558841485x^6+176507096x^7-1712965710x^8-1638575932x^9+33145995x^{10}+32005056x^{11}+14444480x^{12}-451580x^{13}-227805x^{14}-42296x^{15}+1530x^{16}+524x^{17}+5x^{18}\right)\right)/\left(5(-5+x)(-4+x)(-3+x)(-2+x)(-1+x)x(1+x)\left(3+2x+x^2\right)\left(8+3x+x^2\right)\left(15+4x+x^2\right)\left(24+5x+x^2\right)\left(35+6x+x^2\right)\right)\Big\}$$

In[38]:=
```
Clear[DxH];
DxH[n_, x_] := DxH[n, x] =
  Which[n == 1, 2 / 15 * (15 * x^3 + 262 * x^2 + 115 * x - 588) / x / (x + 1),
   n == 2,
```

```
4 / 15 / x * (15 * x^6 + 524 * x^5 + 490 * x^4 - 4910 * x^3 - 3649 * x^2 + 1266 * x + 16 056) /
     (x + 1) / (x - 1) / (x^2 + 2 * x + 3),
True,
Together[
 (* This is the conjectured recurrence that D_p(x) appears to satisfy. *)
 (2 * (100 * n^12 - 26 * n^11 * x - 351 * n^9 * x^3 + 78 * n^8 * x^4 + 453 * n^6 * x^6 -
       78 * n^5 * x^7 - 199 * n^3 * x^9 + 26 * n^2 * x^10 - 3 * x^12 - 1200 * n^11 +
       286 * n^10 * x + 3159 * n^8 * x^3 - 624 * n^7 * x^4 - 2718 * n^5 * x^6 +
       390 * n^4 * x^7 + 597 * n^2 * x^9 - 52 * n * x^10 + 5900 * n^10 -
       897 * n^9 * x - 78 * n^8 * x^2 - 11 730 * n^7 * x^3 + 1122 * n^6 * x^4 +
       156 * n^5 * x^5 + 6642 * n^4 * x^6 - 183 * n^3 * x^7 - 78 * n^2 * x^8 -
       380 * n * x^9 + 12 * x^10 - 15 000 * n^9 - 507 * n^8 * x + 624 * n^7 * x^2 +
       23 142 * n^6 * x^3 + 2004 * n^5 * x^4 - 780 * n^4 * x^5 - 8448 * n^3 * x^6 -
       1011 * n^2 * x^7 + 156 * n * x^8 - 18 * x^9 + 19 500 * n^8 + 9312 * n^7 * x -
       1575 * n^6 * x^2 - 26 037 * n^5 * x^3 - 10 086 * n^4 * x^4 + 963 * n^3 * x^5 +
       5655 * n^2 * x^6 + 828 * n * x^7 - 18 * x^8 - 7200 * n^7 - 23 688 * n^6 * x +
       714 * n^5 * x^2 + 16 701 * n^4 * x^3 + 15 336 * n^3 * x^4 + 231 * n^2 * x^5 -
       1662 * n * x^6 + 54 * x^7 - 13 900 * n^6 + 29 027 * n^5 * x + 3444 * n^4 * x^2 -
       5741 * n^3 * x^3 - 10 868 * n^2 * x^4 - 516 * n * x^5 + 12 * x^6 + 21 000 * n^5 -
       18 703 * n^4 * x - 6888 * n^3 * x^2 + 771 * n^2 * x^3 + 3064 * n * x^4 -
       54 * x^5 - 11 600 * n^4 + 5784 * n^3 * x + 5265 * n^2 * x^2 + 68 * n * x^3 -
       3 * x^4 + 2400 * n^3 - 588 * n^2 * x - 1506 * n * x^2 + 18 * x^3) *
    n / (n - 1) / (n - 1 - x) / (n^2 + n * x + x^2 - 1) /
     (100 * n^9 - 26 * n^8 * x - 251 * n^6 * x^3 + 52 * n^5 * x^4 + 202 * n^3 * x^6 -
       26 * n^2 * x^7 + 3 * x^9 - 1350 * n^8 + 312 * n^7 * x + 2259 * n^5 * x^3 -
       390 * n^4 * x^4 - 909 * n^2 * x^6 + 78 * n * x^7 + 7800 * n^7 - 1309 * n^6 * x -
       52 * n^5 * x^2 - 8231 * n^4 * x^3 + 740 * n^3 * x^4 + 52 * n^2 * x^5 +
       1313 * n * x^6 - 61 * x^7 - 25 200 * n^6 + 1953 * n^5 * x + 390 * n^4 * x^2 +
       15 501 * n^3 * x^3 + 180 * n^2 * x^4 - 156 * n * x^5 - 606 * x^6 + 49 800 * n^5 +
       1601 * n^4 * x - 942 * n^3 * x^2 - 15 916 * n^2 * x^3 - 1482 * n * x^4 +
       113 * x^5 - 61 650 * n^4 - 9417 * n^3 * x + 729 * n^2 * x^2 + 8490 * n * x^3 +
       900 * x^4 + 46 700 * n^3 + 12 874 * n^2 * x + 169 * n * x^2 - 1855 * x^3 -
       19 800 * n^2 - 7788 * n * x - 294 * x^2 + 3600 * n + 1800 * x) * DxH[n - 1, x] -
   (n^2 + n * x + x^2 - 4 * n - 2 * x + 3) * (100 * n^9 - 26 * n^8 * x - 251 * n^6 * x^3 +
       52 * n^5 * x^4 + 202 * n^3 * x^6 - 26 * n^2 * x^7 + 3 * x^9 - 450 * n^8 +
       104 * n^7 * x + 753 * n^5 * x^3 - 130 * n^4 * x^4 - 303 * n^2 * x^6 + 26 * n * x^7 +
       600 * n^7 + 147 * n^6 * x - 52 * n^5 * x^2 - 701 * n^4 * x^3 - 300 * n^3 * x^4 +
       52 * n^2 * x^5 + 101 * n * x^6 - 9 * x^7 - 805 * n^5 * x + 130 * n^4 * x^2 +
       147 * n^3 * x^3 + 580 * n^2 * x^4 - 52 * n * x^5 - 600 * n^5 + 831 * n^4 * x +
       98 * n^3 * x^2 + 26 * n^2 * x^3 - 202 * n * x^4 + 9 * x^5 + 450 * n^4 - 199 * n^3 * x -
       277 * n^2 * x^2 + 26 * n * x^3 - 100 * n^3 - 52 * n^2 * x + 101 * n * x^2 - 3 * x^3) *
    n / (n - 2) / (100 * n^9 - 26 * n^8 * x - 251 * n^6 * x^3 + 52 * n^5 * x^4 +
        202 * n^3 * x^6 - 26 * n^2 * x^7 + 3 * x^9 - 1350 * n^8 + 312 * n^7 * x +
```

```
2259 * n^5 * x^3 - 390 * n^4 * x^4 - 909 * n^2 * x^6 + 78 * n * x^7 + 7800 * n^7 -
1309 * n^6 * x - 52 * n^5 * x^2 - 8231 * n^4 * x^3 + 740 * n^3 * x^4 +
52 * n^2 * x^5 + 1313 * n * x^6 - 61 * x^7 - 25 200 * n^6 + 1953 * n^5 * x +
390 * n^4 * x^2 + 15 501 * n^3 * x^3 + 180 * n^2 * x^4 - 156 * n * x^5 -
606 * x^6 + 49 800 * n^5 + 1601 * n^4 * x - 942 * n^3 * x^2 - 15 916 * n^2 * x^3 -
1482 * n * x^4 + 113 * x^5 - 61 650 * n^4 - 9417 * n^3 * x + 729 * n^2 * x^2 +
8490 * n * x^3 + 900 * x^4 + 46 700 * n^3 + 12 874 * n^2 * x + 169 * n * x^2 -
1855 * x^3 - 19 800 * n^2 - 7788 * n * x - 294 * x^2 + 3600 * n + 1800 * x) /
(n^2 + n * x + x^2 - 1) * DxH[n - 2, x])]];
(* Check that the two sequences, the original one and the one
 defined by the recurrence, are the same: *)
Table[Together[DxH[n, x] - DxR[n, x]], {n, 6}]
```

Out[40]=
```
{0, 0, 0, 0, 0, 0}
```

In[41]:=
```
annR1 = Annihilator[R1[p], S[p]]
```

Out[41]=

$$\{(-232 p^4 - 232 p^5 - 2018 p^6 - 1960 p^7 + 1702 p^8 + 2192 p^9 + 548 p^{10} - 292 p^3 x - 408 p^4 x - 1351 p^5 x - 2200 p^6 x + 3069 p^7 x + 4960 p^8 x + 1514 p^9 x - 118 p^2 x^2 - 438 p^3 x^2 + 1386 p^4 x^2 - 490 p^5 x^2 + 2007 p^6 x^2 + 6348 p^7 x^2 + 2618 p^8 x^2 - 15 p x^3 - 206 p^2 x^3 + 1484 p^3 x^3 + 2260 p^4 x^3 - 383 p^5 x^3 + 2804 p^6 x^3 + 2219 p^7 x^3 - 30 p x^4 + 367 p^2 x^4 + 2560 p^3 x^4 - 1653 p^4 x^4 - 1886 p^5 x^4 + 707 p^6 x^4 + 15 p x^5 + 850 p^2 x^5 - 630 p^3 x^5 - 3888 p^4 x^5 - 1096 p^5 x^5 + 60 p x^6 + 58 p^2 x^6 - 2566 p^3 x^6 - 1537 p^4 x^6 + 15 p x^7 - 644 p^2 x^7 - 1006 p^3 x^7 - 30 p x^8 - 307 p^2 x^8 - 15 p x^9) S_p + (-1212 p - 5222 p^2 - 6576 p^3 + 2714 p^4 + 13500 p^5 + 9690 p^6 - 2424 p^7 - 6634 p^8 - 3288 p^9 - 548 p^{10} - 588 x - 2630 p x - 1135 p^2 x + 11339 p^3 x + 22368 p^4 x + 10698 p^5 x - 11979 p^6 x - 17893 p^7 x - 8666 p^8 x - 1514 p^9 x - 473 x^2 + 2378 p x^2 + 15487 p^2 x^2 + 24550 p^3 x^2 + 4979 p^4 x^2 - 25832 p^5 x^2 - 30875 p^6 x^2 - 14596 p^7 x^2 - 2618 p^8 x^2 + 1553 x^3 + 7397 p x^3 + 8193 p^2 x^3 - 10199 p^3 x^3 - 31430 p^4 x^3 - 29392 p^5 x^3 - 12729 p^6 x^3 - 2219 p^7 x^3 + 1223 x^4 - 144 p x^4 - 12234 p^2 x^4 - 23828 p^3 x^4 - 18382 p^4 x^4 - 6128 p^5 x^4 - 707 p^6 x^4 - 1327 x^5 - 6497 p x^5 - 9628 p^2 x^5 - 3962 p^3 x^5 + 1592 p^4 x^5 + 1096 p^5 x^5 - 1027 x^6 - 1606 p x^6 + 1466 p^2 x^6 + 3582 p^3 x^6 + 1537 p^4 x^6 + 347 x^7 + 1715 p x^7 + 2374 p^2 x^7 + 1006 p^3 x^7 + 277 x^8 + 584 p x^8 + 307 p^2 x^8 + 15 x^9 + 15 p x^9)\}$$

In[42]:=
```
annSum1 = Annihilator[Sum[R2[i, p], {i, 1, p - 1}], S[p]]
```

Out[42]=
$$\{(600 p^4 - 450 p^6 - 150 p^7 - 600 p^8 + 450 p^{10} + 150 p^{11} + 292 p^3 x + 300 p^4 x - 797 p^5 x - 223 p^6 x - 671 p^7 x - 193 p^8 x + 894 p^9 x + 398 p^{10} x - 158 p^2 x^2 + 596 p^3 x^2 - 790 p^4 x^2 - 510 p^5 x^2 - 341 p^6 x^2 - 677 p^7 x^2 + 1136 p^8 x^2 + 744 p^9 x^2 - 77 p x^3 - 85 p^2 x^3 + 312 p^3 x^3 - 714 p^4 x^3 - 219 p^5 x^3 - 657 p^6 x^3 + 659 p^7 x^3 + 781 p^8 x^3 - 154 p x^4 + 503 p^2 x^4 - 281 p^3 x^4 - 477 p^4 x^4 - 493 p^5 x^4 + 237 p^6 x^4 + 665 p^7 x^4 + 25 p x^5 + 477 p^2 x^5 - 676 p^3 x^5 - 326 p^4 x^5 + 122 p^5 x^5 + 378 p^6 x^5 + 204 p x^6 - 198 p^2 x^6 - 396 p^3 x^6 + 131 p^4 x^6 + 259 p^5 x^6 + 77 p x^7 - 313 p^2 x^7 + 84 p^3 x^7 + 152 p^4 x^7 - 50 p x^8 - 43 p^2 x^8 + 93 p^3 x^8 - 25 p x^9 + 25 p^2 x^9) S_p + (-1200 p - 4800 p^2 - 5700 p^3 + 2100 p^4 + 10950 p^5 + 9000 p^6 - 150 p^7 - 5100 p^8 - 3750 p^9 - 1200 p^{10} - 150 p^{11} - 564 x - 2058 p x + 38 p^2 x + 10241 p^3 x + 18149 p^4 x + 9606 p^5 x - 7554 p^6 x - 14703 p^7 x - 9671 p^8 x - 3086 p^9 x - 398 p^{10} x - 306 x^2 + 2371 p x^2 + 12540 p^2 x^2 + 19339 p^3 x^2 + 6116 p^4 x^2 - 17447 p^5 x^2 - 26290 p^6 x^2 - 17019 p^7 x^2 - 5560 p^8 x^2 - 744 p^9 x^2 + 1350 x^3 + 5097 p x^3 + 4941 p^2 x^3 - 6553 p^3 x^3 - 22131 p^4 x^3 - 26174 p^5 x^3 - 16598 p^6 x^3 - 5589 p^7 x^3 - 781 p^8 x^3 + 526 x^4 - 673 p x^4 - 6996 p^2 x^4 - 15232 p^3 x^4 - 17732 p^4 x^4 - 12050 p^5 x^4 - 4418 p^6 x^4 - 665 p^7 x^4 - 1058 x^5 - 3311 p x^5 - 4999 p^2 x^5 - 5712 p^3 x^5 - 4734 p^4 x^5 - 2146 p^5 x^5 - 378 p^6 x^5 - 134 x^6 - 183 p x^6 - 814 p^2 x^6 - 1670 p^3 x^6 - 1164 p^4 x^6 - 259 p^5 x^6 + 322 x^7 + 347 p x^7 - 347 p^2 x^7 - 524 p^3 x^7 - 152 p^4 x^7 - 86 x^8 - 315 p x^8 - 322 p^2 x^8 - 93 p^3 x^8 - 50 x^9 - 75 p x^9 - 25 p^2 x^9)\}$$

In[43]:=
```
(* Convert the second-order recurrence into an operator *)
annDp = {NormalizeCoefficients[
    ToOrePolynomial[f[p, x] == (DxH[n, x] /. DxH → f /. n → p)[[6, 1]], f[p, x]]]};
Support[
 annDp]
```

Out[44]= $\{\{S_p^2, S_p, 1\}\}$

In[45]:=
```
(* Prepare the bivariate summand for the second sum *)
annSmnd = DFiniteTimes[
   ToOrePolynomial[Append[annDp /. p → i, S[p] - 1], OreAlgebra[S[i], S[p]]],
   Annihilator[R3[i, p], {S[i], S[p]}]];
ByteCount[annSmnd]
```

Out[46]= 223 648

In[47]:=
```
ct = CreativeTelescoping[annSmnd, S[i] - 1]
```

Out[47]=
$$\{\{1\}, \{(((2500 i^3 + 3750 i^4 - 1750 i^5 - 4500 i^6 + 2250 i^8 - 750 i^9 - 2250 i^{10} - 250 i^{11} + 750 i^{12} + 250 i^{13} - 3600 i^3 p - 5200 i^4 p + 2100 i^5 p + 4300 i^6 p + 1200 i^8 p + 2700 i^9 p + 100 i^{10} p - 1200 i^{11} p - 400 i^{12} p - 1250 i^2 x - 10902 i^3 x - 12364 i^4 x + 5247 i^5 x + 9651 i^6 x - 516 i^8 x + 2139 i^9 x + 807 i^{10} x - 234 i^{11} x - 78 i^{12} x + 1800 i^2 p x + 4200 i^3 p x + 1650 i^4 p x + 2630 i^5 p x + 5820 i^6 p x - 4260 i^8 p x - 1560 i^9 p x + 390 i^{10} p x + 130 i^{11} p x - 1250 i x^2 + 3576 i^2 x^2 + 12555 i^3 x^2 + 9057 i^4 x^2 - 1671 i^5 x^2 - 1671 i^6 x^2 - 234 i^8 x^2 - 156 i^9 x^2 + 1800 i p x^2 - 60 i^2 p x^2 + 2445 i^3 p x^2 + 3045 i^4 p x^2 - 3075 i^5 p x^2 - 4005 i^6 p x^2 + 390 i^8 p x^2 + 3576 i x^3 + 541 i^2 x^3 + 7794 i^3 x^3 + 9601 i^4 x^3 - 4545 i^5 x^3 -$$

$8403\, i^6 x^3 + 750\, i^8 x^3 - 1515\, i^9 x^3 - 755\, i^{10} x^3 + 265\, i\, p\, x^3 - 2625\, i^2 p\, x^3 - 7840\, i^3 p\, x^3 -$
$3990\, i^4 p\, x^3 - 1590\, i^5 p\, x^3 - 4260\, i^6 p\, x^3 + 3480\, i^8 p\, x^3 + 1560\, i^9 p\, x^3 + 1828\, i\, x^4 -$
$6606\, i^2 x^4 - 23\,160\, i^3 x^4 - 16\,632\, i^4 x^4 + 3186\, i^5 x^4 + 3186\, i^6 x^4 + 234\, i^8 x^4 + 156\, i^9 x^4 -$
$15\, p\, x^4 - 4380\, i\, p\, x^4 + 640\, i^2 p\, x^4 - 2160\, i^3 p\, x^4 - 3360\, i^4 p\, x^4 + 4590\, i^5 p\, x^4 +$
$5670\, i^6 p\, x^4 - 390\, i^8 p\, x^4 - 10\,182\, i\, x^5 + 1893\, i^2 x^5 + 5502\, i^3 x^5 + 3453\, i^4 x^5 +$
$156\, i^6 x^5 - 405\, i\, p\, x^5 + 1695\, i^2 p\, x^5 + 6240\, i^3 p\, x^5 + 2340\, i^4 p\, x^5 - 780\, i^5 p\, x^5 +$
$1256\, i\, x^6 + 3030\, i^2 x^6 + 10\,605\, i^3 x^6 + 7575\, i^4 x^6 - 1515\, i^5 x^6 - 1515\, i^6 x^6 + 60\, p\, x^6 +$
$4140\, i\, p\, x^6 - 840\, i^2 p\, x^6 - 285\, i^3 p\, x^6 + 315\, i^4 p\, x^6 - 1515\, i^5 p\, x^6 - 1665\, i^6 p\, x^6 +$
$9636\, i\, x^7 - 1659\, i^2 x^7 - 4956\, i^3 x^7 - 3219\, i^4 x^7 - 78\, i^6 x^7 + 15\, i\, p\, x^7 - 915\, i^2 p\, x^7 -$
$3900\, i^3 p\, x^7 - 1170\, i^4 p\, x^7 + 390\, i^5 p\, x^7 - 3486\, i\, x^8 - 90\, p\, x^8 - 2340\, i\, p\, x^8 +$
$390\, i^2 p\, x^8 - 3030\, i\, x^9 + 475\, i^2 x^9 + 1470\, i^3 x^9 + 995\, i^4 x^9 + 125\, i\, p\, x^9 + 45\, i^2 p\, x^9 +$
$520\, i^3 p\, x^9 + 2142\, i\, x^{10} + 60\, p\, x^{10} + 780\, i\, p\, x^{10} - 130\, i^2 p\, x^{10} - 490\, i\, x^{12} - 15\, p\, x^{12}\big) \big/$
$\big(60\, (1 + i)\, (1 + i - p)\, \big(-100\, i^3 - 450\, i^4 - 600\, i^5 + 600\, i^7 + 450\, i^8 + 100\, i^9 - 52\, i^2 x +$
$199\, i^3 x + 831\, i^4 x + 805\, i^5 x + 147\, i^6 x - 104\, i^7 x - 26\, i^8 x + 101\, i\, x^2 + 277\, i^2 x^2 +$
$98\, i^3 x^2 - 130\, i^4 x^2 - 52\, i^5 x^2 - 3\, x^3 - 26\, i\, x^3 + 26\, i^2 x^3 - 147\, i^3 x^3 - 701\, i^4 x^3 -$
$753\, i^5 x^3 - 251\, i^6 x^3 - 202\, i\, x^4 - 580\, i^2 x^4 - 300\, i^3 x^4 + 130\, i^4 x^4 + 52\, i^5 x^4 + 9\, x^5 +$
$52\, i\, x^5 + 52\, i^2 x^5 + 101\, i\, x^6 + 303\, i^2 x^6 + 202\, i^3 x^6 - 9\, x^7 - 26\, i\, x^7 - 26\, i^2 x^7 + 3\, x^9\big)\big)\big)$

$S_i + \big(-2500\, i^2 - 6250\, i^3 - 500\, i^4 + 10\,000\, i^5 + 1500\, i^6 - 13\,500\, i^7 - 1500\, i^8 +$
$14\,250\, i^9 + 5500\, i^{10} - 4250\, i^{11} - 2500\, i^{12} - 250\, i^{13} + 3600\, i^2 p +$
$7800\, i^3 p + 1600\, i^4 p - 900\, i^5 p + 4700\, i^6 p - 7200\, i^7 p - 17\,400\, i^8 p -$
$3300\, i^9 p + 7100\, i^{10} p + 3600\, i^{11} p + 400\, i^{12} p + 1250\, i\, x + 12\,152\, i^2 x +$
$24\,046\, i^3 x + 2572\, i^4 x - 20\,613\, i^5 x + 219\, i^6 x + 9996\, i^7 x -$
$7728\, i^8 x - 7881\, i^9 x + 207\, i^{10} x + 702\, i^{11} x + 78\, i^{12} x - 1800\, i\, p\, x -$
$7020\, i^2 p\, x - 4550\, i^3 p\, x - 5980\, i^4 p\, x - 21\,350\, i^5 p\, x - 8880\, i^6 p\, x +$
$17\,160\, i^7 p\, x + 12\,120\, i^8 p\, x - 130\, i^9 p\, x - 1040\, i^{10} p\, x - 130\, i^{11} p\, x +$
$1250\, x^2 - 2326\, i\, x^2 - 17\,646\, i^2 x^2 - 22\,737\, i^3 x^2 - 2061\, i^4 x^2 + 4077\, i^5 x^2 -$
$2919\, i^6 x^2 + 624\, i^7 x^2 + 1170\, i^8 x^2 + 156\, i^9 x^2 - 1800\, p\, x^2 - 990\, i\, p\, x^2 -$
$1650\, i^2 p\, x^2 - 6645\, i^3 p\, x^2 + 5355\, i^4 p\, x^2 + 13\,155\, i^5 p\, x^2 + 2445\, i^6 p\, x^2 -$
$1560\, i^7 p\, x^2 - 390\, i^8 p\, x^2 - 3576\, x^3 - 4072\, i\, x^3 - 8035\, i^2 x^3 - 18\,520\, i^3 x^3 -$
$1681\, i^4 x^3 + 18\,663\, i^5 x^3 + 873\, i^6 x^3 - 9060\, i^7 x^3 + 4530\, i^8 x^3 + 6035\, i^9 x^3 +$
$755\, i^{10} x^3 + 210\, p\, x^3 + 3575\, i\, p\, x^3 + 12\,435\, i^2 p\, x^3 + 10\,660\, i^3 p\, x^3 +$
$6240\, i^4 p\, x^3 + 16\,410\, i^5 p\, x^3 + 7320\, i^6 p\, x^3 - 13\,260\, i^7 p\, x^3 - 10\,560\, i^8 p\, x^3 -$
$1560\, i^9 p\, x^3 - 1828\, x^4 + 4778\, i\, x^4 + 32\,796\, i^2 x^4 + 42\,432\, i^3 x^4 + 3576\, i^4 x^4 -$
$8622\, i^5 x^4 + 4434\, i^6 x^4 - 624\, i^7 x^4 - 1170\, i^8 x^4 - 156\, i^9 x^4 + 4265\, p\, x^4 +$
$2110\, i\, p\, x^4 + 50\, i^2 p\, x^4 + 8220\, i^3 p\, x^4 - 10\,320\, i^4 p\, x^4 - 21\,630\, i^5 p\, x^4 -$
$4110\, i^6 p\, x^4 + 1560\, i^7 p\, x^4 + 390\, i^8 p\, x^4 + 10\,182\, x^5 + 8154\, i\, x^5 -$
$7905\, i^2 x^5 - 8310\, i^3 x^5 - 2673\, i^4 x^5 - 936\, i^5 x^5 - 156\, i^6 x^5 - 1020\, p\, x^5 -$
$5325\, i\, p\, x^5 - 10\,785\, i^2 p\, x^5 - 7800\, i^3 p\, x^5 + 780\, i^5 p\, x^5 - 1256\, x^6 -$
$4286\, i\, x^6 - 15\,150\, i^2 x^6 - 19\,695\, i^3 x^6 - 1515\, i^4 x^6 + 4545\, i^5 x^6 - 1515\, i^6 x^6 -$
$3810\, p\, x^6 - 2160\, i\, p\, x^6 + 1860\, i^2 p\, x^6 - 1575\, i^3 p\, x^6 + 4965\, i^4 p\, x^6 +$
$8475\, i^5 p\, x^6 + 1665\, i^6 p\, x^6 - 9636\, x^7 - 7842\, i\, x^7 + 6735\, i^2 x^7 + 7920\, i^3 x^7 +$
$2829\, i^4 x^7 + 468\, i^5 x^7 + 78\, i^6 x^7 + 1410\, p\, x^7 + 5325\, i\, p\, x^7 + 7665\, i^2 p\, x^7 +$
$4680\, i^3 p\, x^7 - 390\, i^5 p\, x^7 + 3486\, x^8 + 3486\, i\, x^8 + 2040\, p\, x^8 + 1560\, i\, p\, x^8 -$

$$390\, i^2\, p\, x^8 + 3030\, x^9 + 2510\, i\, x^9 - 1855\, i^2\, x^9 - 2510\, i^3\, x^9 - 995\, i^4\, x^9 - 600\, p\, x^9 - 1775\, i\, p\, x^9 - 1515\, i^2\, p\, x^9 - 520\, i^3\, p\, x^9 - 2142\, x^{10} - 2142\, i\, x^{10} - 710\, p\, x^{10} - 520\, i\, p\, x^{10} + 130\, i^2\, p\, x^{10} + 490\, x^{12} + 490\, i\, x^{12} + 15\, p\, x^{12}\Big) \Big/ \Big(60\, i\, (i - p)\, \big(-100\, i^3 - 450\, i^4 - 600\, i^5 + 600\, i^7 + 450\, i^8 + 100\, i^9 - 52\, i^2\, x + 199\, i^3\, x + 831\, i^4\, x + 805\, i^5\, x + 147\, i^6\, x - 104\, i^7\, x - 26\, i^8\, x + 101\, i\, x^2 + 277\, i^2\, x^2 + 98\, i^3\, x^2 - 130\, i^4\, x^2 - 52\, i^5\, x^2 - 3\, x^3 - 26\, i\, x^3 + 26\, i^2\, x^3 - 147\, i^3\, x^3 - 701\, i^4\, x^3 - 753\, i^5\, x^3 - 251\, i^6\, x^3 - 202\, i\, x^4 - 580\, i^2\, x^4 - 300\, i^3\, x^4 + 130\, i^4\, x^4 + 52\, i^5\, x^4 + 9\, x^5 + 52\, i\, x^5 + 52\, i^2\, x^5 + 101\, i\, x^6 + 303\, i^2\, x^6 + 202\, i^3\, x^6 - 9\, x^7 - 26\, i\, x^7 - 26\, i^2\, x^7 + 3\, x^9\big)\Big)\Big\}\Big\}$$

In[48]:=
```
(* Telescoper is 1! Hence the following
  is an antidifference (i.e., an annihilator for it): *)
cert = DFiniteOreAction[annSmnd, ct[[2, 1]]];
ByteCount[cert]
```

Out[49]= 295 600

In[50]:=
```
(* Evaluate the antidifference at the upper summation bound: *)
annUpper = DFiniteSubstitute[cert, {i → p}]
```

Out[50]=

$$\Big\{\Big(3750\, p^2 + 15\,650\, p^3 + 16\,500\, p^4 - 19\,350\, p^5 - 61\,750\, p^6 - 60\,750\, p^7 - 27\,300\, p^8 + 600\, p^9 + 11\,400\, p^{10} + 10\,000\, p^{11} + 4500\, p^{12} + 1050\, p^{13} + 100\, p^{14} + 1875\, p\, x - 5728\, p^2\, x - 65\,923\, p^3\, x - 176\,865\, p^4\, x - 225\,026\, p^5\, x - 142\,039\, p^6\, x - 26\,320\, p^7\, x + 18\,995\, p^8\, x + 10\,628\, p^9\, x - 409\, p^{10}\, x - 1731\, p^{11}\, x - 490\, p^{12}\, x - 39\, p^{13}\, x - 5989\, p\, x^2 - 14\,665\, p^2\, x^2 + 10\,965\, p^3\, x^2 + 76\,649\, p^4\, x^2 + 110\,436\, p^5\, x^2 + 77\,017\, p^6\, x^2 + 28\,013\, p^7\, x^2 + 3657\, p^8\, x^2 - 1178\, p^9\, x^2 - 472\, p^{10}\, x^2 - 61\, p^{11}\, x^2 + 296\, p\, x^3 + 10\,764\, p^2\, x^3 + 64\,210\, p^3\, x^3 + 154\,524\, p^4\, x^3 + 183\,949\, p^5\, x^3 + 105\,323\, p^6\, x^3 + 8619\, p^7\, x^3 - 26\,496\, p^8\, x^3 - 17\,410\, p^9\, x^3 - 5056\, p^{10}\, x^3 - 605\, p^{11}\, x^3 + 11\,361\, p\, x^4 + 30\,950\, p^2\, x^4 + 9250\, p^3\, x^4 - 70\,288\, p^4\, x^4 - 124\,624\, p^5\, x^4 - 95\,172\, p^6\, x^4 - 36\,072\, p^7\, x^4 - 4914\, p^8\, x^4 + 856\, p^9\, x^4 + 195\, p^{10}\, x^4 - 3977\, p\, x^5 - 16\,224\, p^2\, x^5 - 23\,299\, p^3\, x^5 - 11\,988\, p^4\, x^5 + 4933\, p^5\, x^5 + 8164\, p^6\, x^5 + 2644\, p^7\, x^5 + 349\, p^8\, x^5 - 6642\, p\, x^6 - 16\,627\, p^2\, x^6 - 13\,934\, p^3\, x^6 + 9385\, p^4\, x^6 + 30\,669\, p^5\, x^6 + 25\,259\, p^6\, x^6 + 9439\, p^7\, x^6 + 1515\, p^8\, x^6 + 1607\, p\, x^7 + 9617\, p^2\, x^7 + 14\,866\, p^3\, x^7 + 9621\, p^4\, x^7 + 2535\, p^5\, x^7 - 386\, p^6\, x^7 - 156\, p^7\, x^7 + 1606\, p\, x^8 + 275\, p^2\, x^8 - 2846\, p^3\, x^8 - 2525\, p^4\, x^8 - 1010\, p^5\, x^8 - 46\, p\, x^9 - 2266\, p^2\, x^9 - 4960\, p^3\, x^9 - 3995\, p^4\, x^9 - 1255\, p^5\, x^9 - 581\, p\, x^{10} - 336\, p^2\, x^{10} + 245\, p^3\, x^{10} + 245\, p\, x^{11} + 245\, p^2\, x^{11} + 245\, p\, x^{12} + 245\, p^2\, x^{12}\Big)\, S_p^2 +$$

$$\Big(-30\,000\, p^2 - 86\,800\, p^3 - 33\,500\, p^4 + 172\,600\, p^5 + 314\,900\, p^6 + 235\,800\, p^7 + 42\,600\, p^8 - 87\,000\, p^9 - 99\,000\, p^{10} - 53\,600\, p^{11} - 16\,600\, p^{12} - 2800\, p^{13} - 200\, p^{14} - 15\,000\, p\, x + 59\,924\, p^2\, x + 389\,956\, p^3\, x + 744\,502\, p^4\, x + 678\,278\, p^5\, x + 249\,000\, p^6\, x - 76\,032\, p^7\, x - 110\,112\, p^8\, x - 34\,548\, p^9\, x + 2706\, p^{10}\, x + 4302\, p^{11}\, x + 1014\, p^{12}\, x + 78\, p^{13}\, x + 40\,412\, p\, x^2 + 115\,006\, p^2\, x^2 + 58\,272\, p^3\, x^2 - 152\,532\, p^4\, x^2 - 276\,936\, p^5\, x^2 - 201\,168\, p^6\, x^2 - 71\,508\, p^7\, x^2 - 8646\, p^8\, x^2 + 1560\, p^9\, x^2 + 390\, p^{10}\, x^2 + 14\,088\, p\, x^3 - 76\,924\, p^2\, x^3 - 379\,124\, p^3\, x^3 - 646\,630\, p^4\, x^3 - 532\,250\, p^5\, x^3 - 144\,246\, p^6\, x^3 + 124\,782\, p^7\, x^3 + 141\,000\, p^8\, x^3 + 61\,380\, p^9\, x^3 + 13\,310\, p^{10}\, x^3 + 1210\, p^{11}\, x^3 - 75\,616\, p\, x^4 - 210\,248\, p^2\, x^4 - 115\,452\, p^3\, x^4 + 224\,412\, p^4\, x^4 + 405\,906\, p^5\, x^4 + 281\,748\, p^6\, x^4 + 95\,748\, p^7\, x^4 + 11\,676\, p^8\, x^4 - 1560\, p^9\, x^4 - 390\, p^{10}\, x^4 + 1644\, p\, x^5 + 76\,038\, p^2\, x^5 + 131\,148\, p^3\, x^5 + 91\,470\, p^4\, x^5 + 25\,470\, p^5\, x^5 + 312\, p^6\, x^5 - 624\, p^7\, x^5 + 38\,872\, p\, x^6 + 97\,076\, p^2\, x^6 + 57\,180\, p^3\, x^6 - 71\,880\, p^4\, x^6 - 128\,970\, p^5\, x^6 - 80\,580\, p^6\, x^6 - 24\,240\, p^7\, x^6 - 3030\, p^8\, x^6 - 552\, p\, x^7 - 61\,764\, p^2\, x^7 - 100\,104\, p^3\, x^7 - 64\,560\, p^4\, x^7 - 16\,500\, p^5\, x^7 - 156\, p^6\, x^7 + 312\, p^7\, x^7 - 6972\, p\, x^8 - 3486\, p^2\, x^8 - 180\, p\, x^9 + 15\,830\, p^2\, x^9 +$$

$$23\,020\,p^3 x^9 + 12\,550\,p^4 x^9 + 2510\,p^5 x^9 + 4284\,p\,x^{10} + 2142\,p^2 x^{10} - 980\,p\,x^{12} - 490\,p^2 x^{12}\big)\,S_p +$$
$$\big(6600 + 41\,600\,p + 77\,450\,p^2 - 33\,850\,p^3 - 338\,200\,p^4 - 530\,050\,p^5 - 314\,050\,p^6 + 122\,550\,p^7 +$$
$$375\,900\,p^8 + 336\,400\,p^9 + 178\,600\,p^{10} + 61\,600\,p^{11} + 13\,600\,p^{12} + 1750\,p^{13} + 100\,p^{14} + 32\,542\,x +$$
$$132\,047\,p\,x + 171\,478\,p^2 x + 36\,941\,p^3 x - 9455\,p^4 x + 255\,170\,p^5 x + 519\,975\,p^6 x + 461\,592\,p^7 x +$$
$$221\,329\,p^8 x + 54\,228\,p^9 x + 2455\,p^{10} x - 2139\,p^{11} x - 524\,p^{12} x - 39\,p^{13} x - 58\,210\,x^2 -$$
$$279\,571\,p\,x^2 - 488\,999\,p^2 x^2 - 285\,449\,p^3 x^2 + 227\,733\,p^4 x^2 + 501\,120\,p^5 x^2 + 383\,475\,p^6 x^2 +$$
$$165\,031\,p^7 x^2 + 43\,821\,p^8 x^2 + 7498\,p^9 x^2 + 870\,p^{10} x^2 + 61\,p^{11} x^2 - 48\,656\,x^3 - 145\,096\,p\,x^3 -$$
$$85\,340\,p^2 x^3 + 165\,306\,p^3 x^3 + 241\,186\,p^4 x^3 - 23\,363\,p^5 x^3 - 298\,565\,p^6 x^3 - 309\,765\,p^7 x^3 -$$
$$163\,284\,p^8 x^3 - 49\,390\,p^9 x^3 - 8254\,p^{10} x^3 - 605\,p^{11} x^3 + 95\,690\,x^4 + 401\,105\,p\,x^4 + 610\,234\,p^2 x^4 +$$
$$309\,514\,p^3 x^4 - 214\,056\,p^4 x^4 - 400\,670\,p^5 x^4 - 247\,344\,p^6 x^4 - 74\,556\,p^7 x^4 - 8622\,p^8 x^4 +$$
$$704\,p^9 x^4 + 195\,p^{10} x^4 + 2450\,x^5 - 57\,567\,p\,x^5 - 230\,558\,p^2 x^5 - 357\,117\,p^3 x^5 - 290\,042\,p^4 x^5 -$$
$$134\,511\,p^5 x^5 - 35\,976\,p^6 x^5 - 5280\,p^7 x^5 - 349\,p^8 x^5 - 48\,944\,x^6 - 189\,354\,p\,x^6 - 250\,825\,p^2 x^6 -$$
$$93\,446\,p^3 x^6 + 91\,875\,p^4 x^6 + 122\,073\,p^5 x^6 + 60\,453\,p^6 x^6 + 14\,801\,p^7 x^6 + 1515\,p^8 x^6 + 15\,042\,x^7 +$$
$$89\,955\,p\,x^7 + 174\,997\,p^2 x^7 + 158\,088\,p^3 x^7 + 72\,429\,p^4 x^7 + 15\,123\,p^5 x^7 + 542\,p^6 x^7 - 156\,p^7 x^7 +$$
$$6516\,x^8 + 24\,556\,p\,x^8 + 34\,521\,p^2 x^8 + 23\,046\,p^3 x^8 + 7575\,p^4 x^8 + 1010\,p^5 x^8 - 888\,x^9 -$$
$$13\,972\,p\,x^9 - 28\,984\,p^2 x^9 - 23\,200\,p^3 x^9 - 8555\,p^4 x^9 - 1255\,p^5 x^9 - 2142\,x^{10} - 3703\,p\,x^{10} -$$
$$1806\,p^2 x^{10} - 245\,p^3 x^{10} - 490\,x^{11} - 735\,p\,x^{11} - 245\,p^2 x^{11} + 490\,x^{12} + 735\,p\,x^{12} + 245\,p^2 x^{12}\big)\big\}$$

In[51]:=
```
(* Evaluate the antidifference at the lower summation bound: *)
annLower = DFiniteSubstitute[cert, {i → 1}]
```

Out[51]=
$$\big\{\big(3\,p + 7\,p^2 + 5\,p^3 + p^4 + 2\,p\,x + 3\,p^2 x + p^3 x + p\,x^2 + p^2 x^2\big)\,S_p^2 +$$
$$\big(-8\,p^2 - 8\,p^3 - 2\,p^4 - 4\,p\,x - 6\,p^2 x - 2\,p^3 x - 4\,p\,x^2 - 2\,p^2 x^2\big)\,S_p +$$
$$\big(-2 - 3\,p + p^2 + 3\,p^3 + p^4 + 2\,p\,x + 3\,p^2 x + p^3 x + 2\,x^2 + 3\,p\,x^2 + p^2 x^2\big)\big\}$$

In[52]:=
```
annSum2 = DFinitePlus[annUpper, annLower];
Support[annSum2]
```

Out[53]= $\{\{S_p^4,\ S_p^3,\ S_p^2,\ S_p,\ 1\}\}$

In[54]:=
```
(* Sanity check: *)
test = ApplyOreOperator[annSum2[[1]], f[p]];
Together[Table[test, {p, 10}] /. f[p_] :→ Sum[R3[i, p] * DxR[i, x], {i, 1, p - 1}]]
```

Out[55]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

In[56]:=
```
(* Putting all parts together, into a single recurrence *)
annTotal = DFinitePlus[annR1, annSum1, annSum2, annDp];
Support[annTotal]
```

Out[57]= $\{\{S_p^6,\ S_p^5,\ S_p^4,\ S_p^3,\ S_p^2,\ S_p,\ 1\}\}$

In[58]:=
```
(* Look at the integer roots of the leading coefficient: no positive ones. *)
(* Thus, initial values for p = 1,...,
6 need to be checked (has already been done above). *)
Cases[p /. Solve[LeadingCoefficient[annTotal[[1]]] == 0, p], _Integer]
```

Out[58]= {-5, -4, -3, -2, -1, 0}